DEPARTMENT OF MATHEMATICS
Statistical Memoir

No 1
1980
Revised 1983

Some actuarial calculations on HP-25 pocket calculator

by

Bjørn Sundt

Kr. 7,-

Preface

This compendium contains programs for computations in life insurance mathematics on the pocket calculator Hewlett-Packard HP-25. The programs may unchanged also be used on Hewlett-Packard HP-33E. It is assumed that mortality follows Gompertz-Makehams mortality law, and that we have a constant rate of interest.

An earlier version of this compendium was issued as "Programmer for minikalkulatoren HP-25 til beregning av forsikringsmatematiske uttrykk under Gompertz - Makenhams dødelighetslov" (Stat. Mem. No. 1, 1977, in Norwegian) and was then written as a further development of E. Sverdrup: "Some actuarial calculations on electronic desk calculator" (Stat. Mem. No. 1 , 1976).

Blindern, February 1980.

Bjørn Sundt

Revised version 1983

Contents

1. Notation

The force of mortality at age x is given by $\mu_x = \alpha+\beta c^x$, where $c > 0$; the annual rate of interest is denoted by i, and the force of interest is $\delta = \ln(1+i)$.

The expected present value of a t year pure endowment of 1 is denoted by ${}_tE_x$, and we have

$${}_tE_x = e^{-\int_0^t(\delta+\mu_{x+s})ds} = e^{-(\alpha+\delta)t-\frac{\beta}{\ln c}c^x(c^t-1)} .$$

The expected present value of an n year temporary annuity-due is given by

$$\ddot{a}_{x:\overline{n}|} = \sum_{t=0}^{n-1} {}_tE_x ;$$

and the expected present value of an n year temporary continuous annuity is

$$\bar{a}_{x:\overline{n}|} = \int_0^n {}_tE_x \, dt .$$

If $n = \infty$, we use the symbols

$$\ddot{a}_x = \sum_{t=0}^{\infty} {}_tE_x$$

$$\bar{a}_x = \int_0^{\infty} {}_tE_x .$$

Fractional life annuities payable m times a year are denoted by

$$\ddot{a}^{(m)}_{x:\overline{n}|} = \frac{1}{m} \sum_{t=0}^{nm-1} {}_{\frac{t}{m}}E_x$$

$$\ddot{a}^{(m)}_x = \frac{1}{m} \sum_{t=0}^{\infty} {}_{\frac{t}{m}}E_x .$$

2. Tabulation of μ_{x+t}

1	STO 1	6	RCL 4	11	STO+1
2	f y^x	7	RCL 5	12	RCL 3
3	STO×4	8	+	13	GTO 03
4	RCL 1	9	R/S		
5	f PAUSE	10	1		

Set: α, STO 5, β, STO 4, c, STO 3, x .

By pressing R/S repeatedly x+t is displayed at pause and μ_{x+t} at stop (t = 0,1,2,...).

Example: α = 0.0025, β = 0.00002, c = 1.11 give for x = 20

x+t	μ_{x+t}
20	0.00266
21	0.00268
22	0.00270

etc.

Registers:

0	unused	4	βc^{x+t}
1	x+t	5	α
2	unused	6	unused
3	c	7	"

3. Single values of ${}_{n}E_{x}$

1	f ln	10	RCL 3	19	×
2	+	11	RCL 2	20	RCL 5
3	STO 5	12	f y^x	21	RCL 2
4	↓	13	-	22	×
5	RCL 3	14	RCL 4	23	-
6	f ln	15	×	24	g e^x
7	÷	16	RCL 3	25	R/S
8	STO 4	17	RCL 1		
9	1	18	f y^x		

Set: x, STO 1, n, STO 2, c, STO 3, β, enter, α, enter, 1+i, R/S .

${}_{n}E_{x}$ is displayed.

Other values of ${}_{n}E_{x}$ on the same standard (i,α,β,c):

Set: x, STO 1, n, STO 2, GTO 09, R/S .

Example: α = 0.0025, β = 0.00002, c = 1.11 , i = 0.03 give

${}_{10}E_{20} = 0.72366$.

Registers:

0	unused	4	$\frac{\beta}{\ln c}$
1	x	5	α+δ
2	n	6	unused
3	c	7	"

4. Main Program I

The program is based on the recursions

$$\ddot{a}_{x+t:\overline{n-t}|} = 1 + {}_1E_{x+t}\ \ddot{a}_{x+t+1:\overline{n-t-1}|} \quad ; \quad \ddot{a}_{x+n:\overline{0}|} = 0$$

$$_{n-t}E_{x+t} = {}_1E_{x+t}\ {}_{n-t-1}E_{x+t+1} \quad ; \quad {}_0E_{x+n} = 1 .$$

4A. Single values of $\ddot{a}_{x:\overline{n}|}$

1	f ln	13	RCL 4	25	-
2	+	14	RCL 3	26	g e^x
3	STO 5	15	RCL 1	27	STO×7
4	↓	16	f y^x	28	1
5	RCL 3	17	×	29	STO+7
6	f ln	18	STO 1	30	STO-2
7	÷	19	0	31	RCL 2
8	1	20	STO 7	32	g x≠0
9	RCL 3	21	RCL 3	33	GTO 21
10	-	22	STO÷1	34	RCL 7
11	×	23	RCL 1	35	R/S
12	STO 4	24	RCL 5		

Set: x+n, STO 1, n, STO 2, c, STO 3, β, enter, α, enter, 1+i, R/S .

$\ddot{a}_{x:\overline{n}|}$ is displayed.

Other values of $\ddot{a}_{x:n}$ on the same standard (i,α,β,c):

Set: x+n, STO 1, n, STO 2, GTO 13, R/S .

<u>Example</u>: α = 0.0025, β = 0.00002, c = 1.11, i = 0.03 give

$\ddot{a}_{20:\overline{10}|}$ = 8.68491 .

Registers:

0	unused	4	$-\frac{\beta}{\ln c}(c-1)$
1	$-\frac{\beta}{\ln c}c^{x+n-t}(c-1)$	5	$\alpha+\delta$
2	n-t	6	unused
3	c	7	$\ddot{a}_{x+n-t:\overline{t}\rceil}$

4B. Single values of $_nE_x$

Make the following changes in the program of subsection 3A :

19 1 29 g NOP

Set: x+n, STO 1, n, STO 2, c, STO 3, β, enter, α, enter, 1+i, R/S .

$_nE_x$ is displayed.

Other values of $_nE_x$ on the same standard (i,α,β,c):

Set: x+n, STO 1, n, STO 2, GTO 13, R/S .

Example: α = 0.0025, β = 0.00002, c = 1.11, i = 0.03 give

$_{10}E_{20}$ = 0.72366

Registers: As in 4A with the following change:

7 $_tE_{x+n-t}$

4C. Tabulation of $\ddot{a}_{x:\overline{w-x}\rceil}$ with respect to x

Make the following changes in the program of subsection 4A:

32 f PAUSE

33 RCL 7

34 R/S

35 GTO 21

Set: w, STO 1, STO 2, c, STO 3, β, enter, α, enter, 1+i .

By pressing R/S repeatedly x is displayed at pause and $\ddot{a}_{x:\overline{w-x|}}$ at stop (x = w-1, w-2, ...).

New w on the same standard (i,α,β,c):

Set: w, STO 1, STO 2, GTO 13.

Example: α = 0.0025, β = 0.00002, c = 1.11, i = 0.03 give

x	$\ddot{a}_{x:\overline{70-x|}}$
69	1.00000
68	1.94410
67	2.84007

etc.

Registers:

0	unused	4	$-\frac{\beta}{\ln c}(c-1)$
1	$-\frac{\beta}{\ln c}c^{x}(c-1)$	5	$\alpha+\delta$
2	w-x	6	unused
3	c	7	$\ddot{a}_{x:\overline{w-x|}}$

4D. Tabulation of ${}_{w-x}E_x$ with respect to x

Make the following changes in the program of subsection 3C:

19	1	29	g NOP

Set: w, STO 1, STO 2, c, STO 3, β, enter, α, enter, 1+i .

By pressing R/S repeatedly x is displayed at pause and ${}_{w-x}E_x$ at stop (x = w-1, w-2, ...).

New w on the same standard (i,α,β,c):

Set: w, STO 1, STO 2, GTO 13.

Example: $\alpha = 0.0025$, $\beta = 0.00002$, $c = 1.11$, $i = 0.03$

give

x	${}_{70-x}E_x$
69	0.94146
68	0.88884
67	0.84128

etc.

Registers: As in 4C with the following change:

7 ${}_{w-x}E_x$

5. Main Program II

The probram is based on the recursions

$$\ddot{a}_{x:\overline{t|}} = \ddot{a}_{x:\overline{t-1|}} + {}_{t-1}E_x \quad ; \quad \ddot{a}_{x:\overline{0|}} = 0$$

$${}_tE_x = {}_1E_{x+t-1}\ {}_{t-1}E_x \quad ; \quad {}_0E_x = 1 \ .$$

5A. Single values of $\ddot{a}_{x:\overline{n|}}$ and ${}_nE_x$

1	f ln	14	RCL 3	27	-
2	+	15	RCL 1	28	g e^x
3	STO 5	16	f y^x	29	STO×6
4	↓	17	×	30	RCL 3
5	RCL 3	18	STO 1	31	STO×1
6	f ln	19	0	32	1
7	÷	20	STO 7	33	STO-2
8	1	21	1	34	RCL 2
9	RCL 3	22	STO 6	35	g x≠0
10	-	23	RCL 6	36	GTO 23
11	×	24	STO+7	37	RCL 7
12	STO 4	25	RCL 1	38	R/S
13	RCL 4	26	RCL 5		

(The 20 first program lines are the same as in Main Program I.)

Set: x, STO 1, n, STO 2, c, STO 3, β, enter, α, enter, 1+i, R/S .

$\ddot{a}_{x:\overline{n}|}$ is displayed.

By RCL 6 $_{n}E_{x}$ is displayed.

Other values of $\ddot{a}_{x:\overline{n}|}$ and $_{n}E_{x}$ on the same standard (i,α,β,c):

Set: x, STO 1, n, STO 2, GTO 13, R/S .

Example: α = 0.0025, β = 0.00002, c = 1.11, i = 0.03 give

$\ddot{a}_{20:\overline{10}|} = 8.68491$ and $_{10}E_{20} = 0.72366$

Registers:

0	unused	4	$-\frac{\beta}{\ln c}(c-1)$
1	$-\frac{\beta}{\ln c}c^{x+t}(c-1)$	5	$\alpha+\delta$
2	n-t	6	$_{t}E_{x}$
3	c	7	$\ddot{a}_{x:\overline{t}|}$

5B. Tabulation of $\ddot{a}_{x:\overline{t}|}$ and $_{t}E_{x}$ with respect to t

Make the following changes in the program of subsection 5A:

33	STO+2	35	f PAUSE
		36	RCL 7
		37	R/S
		38	GTO 23

Set: x, STO 1, 0, STO 2, c, STO 3, β, enter, α, enter, 1+i .

By pressing R/S repeatedly t is displayed at pause, $\ddot{a}_{x:t}$ at stop, and $_{t}E_{x}$ by RCL 6 (t = 1,2,3,...).

New x on the same standard (i,α,β,c):

Set: x, STO 1, 0, STO 2, GTO 13 .

Example: $\alpha = 0.0025$, $\beta = 0.00002$, $c = 1.11$, $i = 0.03$ give

t	$\ddot{a}_{20:\overline{t}|}$	${}_tE_{20}$
1	1.00000	0.96829
2	1.96829	0.93756
3	2.90584	0.90779
	etc.	

Registers: As in 5A with the following change:

2 t

5C. Single values of $\ddot{a}_x$

Make the following changes in the program of subsection 4A:

33 RCL 6

34 RCL 0

35 f x<y

Set: ε, STO 0, x, STO 1, c, STO 3, β, enter, α,enter, 1+i, R/S .

The program computes $\ddot{a}_{x:\overline{n}|}$, where n is the smallest integer satisfying ${}_nE_x \leq \varepsilon$.

Other values of $\ddot{a}_x$ on the same standard (i,α,β,c) and with the same ε:

Set: x, STO 1, GTO 13, R/S .

Example: $\alpha = 0.0025$, $\beta = 0.00002$, $c = 1.11$, $i = 0.03$, $\varepsilon = 10^{-8}$ give

$\ddot{a}_{70} = 9.99513$

Registers: As in 5A with the following changes:

0 ε 2 t

6. Continuous life annuities by Euler-Maclaurin's formula

The programs utilize the approximation

$$\bar{a}_x \approx \ddot{a}_x - \tfrac{1}{2} - \tfrac{1}{12}(\mu_x + \delta)$$

and the recursions described in sections 4 and 5.

6A. Single values of $\bar{a}_{x:\overline{n}|}$

1	f ln	17	STO 1	33	STO×7
2	+	18	+	34	1
3	STO 5	19	6	35	STO÷7
4	1	20	÷	36	STO-2
5	RCL 3	21	1	37	RCL 2
6	-	22	2	38	g x≠0
7	RCL 3	23	÷	39	GTO 25
8	f ln	24	STO 7	40	RCL 7
9	÷	25	RCL 3	41	RCL 5
10	STO 6	26	STO÷1	42	RCL 1
11	RCL 5	27	RCL 1	43	6
12	RCL 4	28	RCL 6	44	+
13	RCL 3	29	×	45	+
14	RCL 1	30	RCL 5	46	1
15	f y^x	31	-	47	2
16	×	32	g e^x	48	÷
				49	-

Set: x+n, STO 1, n, STO 2, c, STO 3, β, STO 4, α, enter, 1+i, R/S.

$\bar{a}_{x:\overline{n}|}$ is displayed.

Other values of $\bar{a}_{x:\overline{n}|}$ on the same standard (i,α,β,c):

Set: x+n, STO 1, n, STO 2, GTO 11, R/S.

Example: α = 0.0025, β = 0.00002, c = 1.11, i = 0.03 give

$\bar{a}_{20:\overline{10}|} = 8.54602.$

Registers:

0	unused	4	β
1	βc^{x+n-t}	5	$\alpha+\delta$
2	$n-t$	6	$-\frac{c-1}{\ln c}$
3	c	7	$\ddot{a}_{x+n-t:\overline{t}\rceil} + [\frac{1}{2}+\frac{1}{12}(\mu_{x+n}+\delta)]\ {}_tE_{x+n-t}$

6B. Tabulation of $\bar{a}_{x:\overline{w-x}\rceil}$ with respect to x

1	f ln	17	STO 1	33	STO+7
2	+	18	+	34	1
3	STO 5	19	6	35	STO-2
4	1	20	+	36	RCL 2
5	RCL 3	21	STO 7	37	f PAUSE
6	-	22	RCL 3	38	RCL 7
7	RCL 3	23	STO÷1	39	RCL 5
8	f ln	24	RCL 1	40	RCL 1
9	÷	25	RCL 6	41	6
10	STO 6	26	×	42	+
11	RCL 5	27	RCL 5	43	+
12	RCL 4	28	-	44	-
13	RCL 3	29	g e^x	45	1
14	RCL 2	30	STO×7	46	2
15	f y^x	31	1	47	÷
16	×	32	2	48	R/S
				49	GTO 22

Set: w, STO 2, c, STO 3, β, STO 4, α, enter, 1+i .

By pressing R/S repeatedly x is displayed at pause and $\bar{a}_{x:\overline{w-x}\rceil}$ at stop (x = w-1, w-2, ...).

New w on the same standard (i, α, β, c):

Set: w, STO 2, GTO 11.

Example: $\alpha = 0.0025$, $\beta = 0.00002$, $c = 1.11$, $i = 0.03$ give

x	$\bar{a}_{x:\overline{70-x}\rceil}$
69	0.97068
68	1.88842
67	2.76056

etc.

Registers:

0	unused	4	β
1	βc^x	5	$\alpha+\delta$
2	x	6	$-\frac{c-1}{\ln c}$
3	c	7	$12\ddot{a}_{x:\overline{w-x}\rceil} + (6+\mu_w+\delta)\,{}_{w-x}E_x$

6C. Tabulation of $\bar{a}_{x:\overline{t}\rceil}$, $\ddot{a}_{x:\overline{t}\rceil}$, and ${}_tE_x$ with respect to t

1	RCL 4
2	RCL 3
3	RCL 1
4	f y^x
5	×
6	STO 0
7	STO 1
8	1
9	STO 2
10	STO 7
11	RCL 3
12	STO×1
13	RCL 0
14	RCL 1
15	-
16	RCL 3
17	f ln
18	÷
19	RCL 2
20	f PAUSE
21	RCL 5
22	×
23	-
24	g e^x
25	STO 6
26	1
27	STO+2
28	-
29	6
30	RCL 5
31	+
32	×
33	RCL 6
34	RCL 1
35	×
36	+
37	RCL 0
38	-
39	1
40	2
41	÷
42	RCL 7
43	+
44	R/S
45	RCL 6
46	STO+7
47	GTO 11

Set: x, STO 1, c, STO 3, β, STO 4, α+δ, STO 5.

By pressing R/S repeatedly t is shown at pause, $\bar{a}_{x:\overline{t|}}$ at stop, ${}_tE_x$ by RCL 6, and $\ddot{a}_{x:\overline{t|}}$ by RCL 7 .

New x on the same standard (i,α,β,c):

Set: x, STO 1, GTO 00, R/S .

<u>Example</u>: α = 0.0025, β = 0.00002, c = 1.11, i = 0.03 give

t	$\bar{a}_{20:\overline{t|}}$	$\ddot{a}_{20:\overline{t|}}$	${}_tE_{20}$
1	0.98406	1.00000	0.96828
2	1.93690	1.96829	0.93756
3	2.85949	2.90584	0.90770

etc.

Registers:

0	βc^x	4	β
1	βc^{x+t}	5	α+δ
2	t	6	${}_tE_x$
3	c	7	$\ddot{a}_{x:\overline{t|}}$

6 D. Single values of $\bar{a}_x$

1	f ln	15	f y^x	29	RCL 1
2	+	16	×	30	RCL 5
3	STO 5	17	STO 1	31	-
4	1	18	RCL 5	32	g e^x
5	RCL 3	19	+	33	STO×6
6	-	20	-	34	RCL 3
7	RCL 3	21	1	35	STO×1
8	f ln	22	2	36	RCL 6
9	÷	23	÷	37	STO+7
10	STO 2	24	STO 7	38	RCL 0
11	6	25	RCL 2	39	f x<y
12	RCL 4	26	STO×1	40	GTO 29
13	RCL 3	27	1	41	RCL 7
14	RCL 1	28	STO 6	42	R/S

Set: ε, STO 0, x, STO 1, c, STO 3, β, STO 4, α, enter, 1+i, R/S.

The program computes $\ddot{a}_{x:\overline{n}|} - \frac{1}{2} - \frac{1}{12}(\mu_x+\delta)$, where n is the smallest integer, such that ${}_{n-1}E_x \leq \varepsilon$.

Other values of $\bar{a}_x$ on the same standard (i,α,β,c) and with the same ε :

Set: x, STO 1, GTO 11, R/S .

Example: $\alpha = 0.0025$, $\beta = 0.00002$, $c = 1.11$, $i = 0.03$, $\varepsilon = 10^{-8}$ give

$$\bar{a}_{70} = 9.48998$$

Registers:

0	ε	4	β
1	$-\frac{\beta}{\ln c} c^{x+t}(c-1)$	5	$\alpha+\delta$
2	$-\frac{c-1}{\ln c}$	6	${}_tE_x$
3	c	7	$\ddot{a}_{x:\overline{t}|} - \frac{1}{2} - \frac{1}{12}(\mu_x+\delta)$

7. Continuous life annuities by Simpsons formula

7A. Single values of $\bar{a}_{x:\overline{n|}}$

```
 1  STO+4
 2   ÷
 3  STO 6
 4  STO 7
 5  RCL 5
 6   ×
 7  STO 0
 8  RCL 3
 9  RCL 1
10  f y^x
11  RCL 3
12  f ln
13   ÷
14   1
15  RCL 3
16  RCL 6
17  f y^x
18  STO 2
19   -
20   ×
21  RCL 4
22  g FRAC
23   ×
24  STO 1
25  RCL 1
26  RCL 0
27   -
28  g e^x
29  CHS
30  STO×6
31   3
32  RCL 6
33  g ABS
34   ×
35  RCL 6
36   -
37  STO+7
38  RCL 2
39  STO×1
40   1
41  STO-4
42  RCL 4
43  f x≥y
44  GTO 25
45  RCL 7
46  RCL 6
47   -
48   3
49   ÷
```

Set: x, STO 1, c, STO 3, β, STO 4, $\alpha+\delta$, STO 5, n, enter, 2m, R/S.

$\bar{a}_{x:\overline{n|}}$ is displayed.

Other values of $\bar{a}_{x:\overline{n}|}$ on the same standard (i,α,β,c):

Set: x, STO 1, n, enter, 2m, R/S .

For an accuracy up to fourth decimal one ought to use $2m \geq \frac{n}{3}$ (that is, $h \leq 3$).

Example: $\alpha = 0.025$, $\beta = 0.00002$, $c = 1.11$, $i = 0.03$ give

2m	$\bar{a}_{20:\overline{60}|}$
2	25.43130238
4	25.30967125
8	25.29246814
16	25.29121787
30	25.29114415
60	25.29113818
100	25.29113784
150	25.29113785

(The program of subsection 6A gives $\bar{a}_{20:\overline{60}|} = 25.29113766$.)

Registers:

0	$h(\alpha+\delta)$	4	$\beta+2m-j$
1	$\frac{\beta}{\ln c} c^{x+jh}(c^h-1)$	5	$\alpha+\delta$
2	c^h	6	$(-1)^j\ h\ {}_{jh}E_x$
3	c	7	Σ

(Because of double storage in register 4 we must have $\beta \in [0,1>$, but values outside this interval are unlikely to appear in practical applications.)

7B. Single values of $\bar{a}_x$

The program uses Simpsons formula with $h = 3$. This gives accuracy up to about fourth decimal.

1	f ln	15	×	29	g e^x
2	+	16	STO 4	30	CHS
3	STO 0	17	RCL 4	31	STO×6
4	↓	18	RCL 3	32	RCL 2
5	RCL 3	19	RCL 1	33	STO×1
6	f ln	20	f y^x	34	3
7	÷	21	×	35	RCL 6
8	1	22	STO 1	36	g ABS
9	RCL 3	23	1	37	×
10	3	24	STO 6	38	RCL 6
11	STO×0	25	STO 7	39	-
12	f y^x	26	RCL 1	40	STO+7
13	STO 2	27	RCL 0	41	g x≠0
14	-	28	-	42	GTO 26
				43	RCL 7
				44	R/S

Set: x, STO 1, c, STO 3, β, enter, α, enter, 1+i, R/S .

$\bar{a}_x$ is displayed.

Other values of $\bar{a}_x$ on the same standard (i,α,β,c):

Set: x, STO 1, GTO 17, R/S.

<u>Example</u>: $\alpha = 0.0025$, $\beta = 0.00002$, $c = 1.11$, $i = 0.03$ give

$\bar{a}_{70} = 9.48998$

Registers:

0	$3(\alpha+\delta)$	3	$-\frac{\beta}{\ln c}(c^3-1)$
1	$-\frac{\beta}{\ln c}c^{x+3j}(c^3-1)$	5	unused
2	c^3	6	$(-1)^j \; {}_{3j}E_x$
3	c	7	$\bar{a}_x$

8. General comments

8A. Fractional life annuities

$\ddot{a}^{(m)}_{x:\overline{n}|} = \frac{1}{m}\ddot{a}'_{xm:\overline{nm}|}$, where $\ddot{a}'_{xm:\overline{nm}|}$ is computed with rate of interest i' and force of mortality $\mu'_x = \alpha' + \beta c'^x$; $1+i' = (1+i)^{\frac{1}{m}}$ $\alpha' = \frac{\alpha}{m}$, $\beta' = \frac{\beta}{m}$, and $c' = c^{\frac{1}{m}}$.

From this follows that fractional annuities may be computed by Main Programs I and II.

8B. Joint-life statuses

We have m independent lives $(x_1),\ldots,(x_m)$ that follow the mortality law $\mu_x = \alpha+\beta\, c^x$. Let

$$w = \frac{\ln \sum_{j=1}^{m} c^{x_j}}{\ln c} .$$

Then the remaining life-time of the joint-life $(x_1,\ldots,x_m)$ has the same distribution as a single life (w) following the mortality law $\mu'_x = m\,\alpha+\beta\, c^x$.

Hence all programs may be used on joint-life statuses.

Appendix. Simpsons formula

A. We want to compute the integral of an integrable function f on a finite interval $[a,b]$.

Let m be an integer and $h = \frac{b-a}{2m}$. Then

$$\int_a^b f(x)dx \approx \frac{h}{3}\{f(a) + \sum_{j=1}^{2m-1} [3-(-1)^j]f(a+jh)+f(b)\} .$$

The accuracy is improved when m is increased.

B. Generalization to intervals of the form $[a,\infty>$.

Let h be a positive number. Then

$$\int_a^\infty f(x)dx \approx \frac{h}{3}\{f(a) + \sum_{j=1}^{\infty} [3-(-1)^j]f(a+jh)\}$$

if both the sum and the integral exist. The accuracy is improved when h decreases.